

УДК 372.851

## Вычислительный эксперимент в преподавании высшей математики. Геометрия

**Куланин Е.Д.\***

Московский государственный психолого-педагогический университет  
(ФГБОУ ВО МГППУ), г. Москва, Российская Федерация  
ORCID: <https://orcid.org/0000-0001-6093-7012>  
e-mail: [lucas03@mail.ru](mailto:lucas03@mail.ru)

**Степанов М.Е.\*\***

Московский государственный психолого-педагогический университет  
(ФГБОУ ВО МГППУ), г. Москва, Российская Федерация  
ORCID: <https://orcid.org/0000-0003-4803-8211>  
e-mail: [mestepanov@yandex.ru](mailto:mestepanov@yandex.ru)

Статья продолжает цикл ([1] – [13]) методических разработок авторов. В ней обсуждаются некоторые проблемы, связанные с путями повышения культуры математического мышления студентов-математиков. Авторы опираются на опыт работы на факультете информационных технологий МГППУ.

**Ключевые слова:** высшее образование, методика преподавания математики, аналитическая геометрия, аффинная геометрия, проективная геометрия, геометрические преобразования, неевклидова геометрия, сферическая геометрия, кривые второго порядка.

**Для цитаты:**

Куланин Е.Д., Степанов М.Е. Вычислительный эксперимент в преподавании высшей математики. Геометрия. // Моделирование и анализ данных. 2024. Том 14. № 4. С. 138–162. DOI: <https://doi.org/10.17759/mda.2024140410>

\***Куланин Евгений Дмитриевич**, кандидат физико-математических наук, профессор, Московский государственный психолого-педагогический университет (ФГБОУ ВО МГППУ), г. Москва, Российская Федерация, ORCID: <https://orcid.org/0000-0001-6093-7012>, e-mail: [lucas03@mail.ru](mailto:lucas03@mail.ru)

\*\***Степанов Михаил Евграфович**, кандидат педагогических наук, доцент, Московский государственный психолого-педагогический университет (ФГБОУ ВО МГППУ), г. Москва, Российская Федерация, ORCID: <https://orcid.org/0000-0003-4803-8211>, e-mail: [mestepanov@yandex.ru](mailto:mestepanov@yandex.ru)



## 1. ВВЕДЕНИЕ

Данная статья в некотором смысле продолжает и дополняет статью авторов «Вычислительный эксперимент. Теория чисел» [13].

Создание Декартом метода координат привело к систематическому внедрению вычислительных методов в геометрию. Сила вычислений оказалась настолько велика, что с этого момента началось фактическое создание математики Нового времени. При этом математики, склонные к традиционной форме геометрии (синтетическая геометрия) часто активно препятствовали триумфальному шествию аналитической геометрии. Однако универсальность и способность приходить к результату без сложных и узко специализированных рассмотрений геометрических проблем привела к тому, что синтетическая геометрия оказалась в роли изгоя.

При этом развитие математики в последнее время демонстрирует некоторое торжество. Возможно, это связано с тем, что при изучении математических теорий упор делается на освоении вычислительных подходов к математическим задачам вообще и геометрии в частности. Более правильным представляется гармоничное сочетание синтетических и аналитических подходов к решению геометрических задач. Размышления и вычисления должны помогать друг другу. По крайней мере, к этому необходимо стремиться в учебном процессе.

## 2. СФЕРИКА

Современные геометрические теории многочисленны и часто неотделимы от вычислительных методов. Однако в данной статье будет рассмотрен круг вопросов, связанных с геометрией на сфере. Этот раздел геометрической науки всегда был востребован в астрономии. С соответствующей тематикой в её традиционной форме можно познакомиться по книге [14]. При этом геометрия на сфере долгое время рассматривалась как раздел евклидовой стереометрии.

Но на границе семнадцатого и восемнадцатого веков стали актуальными новые для того времени исследования, приведшие в итоге к созданию новых геометрических теорий, таких как проективная геометрия и неевклидовы геометрии. Математики постепенно начали понимать, что геометрия на сфере может рассматриваться как особая теория, в чём-то подобная планиметрии Евклида, но без параллельных линий. Более того, Иоганн Ламберт высказал мысль о том, что если бы существовала сфера с отрицательным радиусом, то на ней выполнялась бы теория, которая в дальнейшем будет названа геометрией Лобачевского.

После оформления и признания неевклидовых геометрий и проективной геометрии, многие результаты того времени, связанные со сферической геометрией, были забыты. В частности, речь идёт о теории кривых на сфере, которые являются аналогами кривых второго порядка. Мы рассмотрим некоторые теоремы, связанные с этой тематикой. Но сначала дадим краткие исторические сведения о так называемой **сферике** – учении о геометрических образах, на поверхности шара [15].



Сферика активно разрабатывалась в Петербургской академии наук. Работали в этом направлении академики Леонард Эйлер и Андрей Иванович Лексель, родившийся в Швеции. Но наибольших результатов в этом направлении достиг уроженец Швейцарии Николай Иванович Фусс, который по приглашению Леонарда Эйлера переехал в Россию, чтобы быть его бессменным секретарём. Кроме того, Фусс стал зятем Эйлера. Во времена нашествия Наполеона академик Фусс фактически руководил Академией. Естественно, что большинство его работ было связано с работами Л. Эйлера. Сферика в этом отношении не является исключением. Однако Фусс продвинулся в развитии соответствующей тематики дальше своих предшественников.

В частности, Фусс пришёл к мысли рассмотреть геометрическое место точек, сумма расстояний которых от двух данных точек постоянна, и исследовать ее свойства. Фактически это был шаг к созданию аналитической геометрии сферы. Кроме того, Фусс рассматривал аналоги циклоидальных кривых на сфере, но в данной статье будут рассмотрены только сферические аналоги эллипсов, гипербол и парабол.

Поясним, по каким причинам рассмотрение этих вопросов может представить интерес.

- Сферическая геометрия очень наглядна.
- Сферическая геометрия связана с эллиптической геометрией Римана и проективной геометрией на плоскости.
- Ряд теорем сферической геометрии очень просты, и их доказательство доступно для студентов.
- Некоторые факты сферической геометрии наоборот достаточно сложны и требуют для своего доказательства кропотливых алгебраических преобразований.
- Визуализация на экране объектов сферической геометрии требует овладения сложными вычислительными алгоритмами.

### 3. АНАЛОГИ КРИВЫХ ВТОРОГО ПОРЯДКА НА СФЕРЕ

Начнём с перечисления наиболее важных геометрических объектов на сфере. Прежде всего, следует назвать **плоские сечения** сферы, которые **являются окружностями евклидовой плоскости**. С каждым плоским сечением связан перпендикулярный ему диаметр сферы, который мы будем называть **сопряжённым соответствующему сечению**. Концы любого диаметра являются полюсами сферы. Если секущая плоскость проходит через центр сферы, то сечение называется **линией большого круга**.

Расстояние между двумя точками А и В на сфере можно определить как величину угла АОВ, где О – центр сферы. Фактически та же метрика (с точностью до множителя) будет получена, если расстоянием считать длину дуги линии большого круга, соединяющей точки А и В. При таком определении расстояния кратчайшими линиями, соединяющей две точки на сфере, являются линии большого круга. Таким образом, **роль прямых на сфере играют линии большого круга**. **Остальные плоские сечения сферы являются окружностями**, центрами которых являются полюса сопряжённого сечению диаметра.



Расстояние на сфере между концами любого диаметра постоянно и равно длине меридиана данной сферы. Обозначим эту длину через  $m$ .

Рассматривая сферу как метрическое пространство, можно определить на сфере **аналоги кривых второго порядка**.

**Эллипс** – это множество точек на сфере, сумма расстояний от каждой из которых до двух фиксированных точек, называемых фокусами, постоянна.

**Гипербола** – это множество точек на сфере, разность расстояний от каждой из которых до двух фиксированных точек, называемых фокусами, постоянна.

**Парабола** – это множество точек на сфере, расстояния от которой до фиксированной точки, называемой фокусом, и до сферической прямой, называемой директрисой, одинаковы.

Сразу следует указать на особенность этого метрического пространства. Если точки  $A$  и  $B$  являются концами диаметра сферы, то сумма расстояний от них до любой точки на сфере постоянна и равна длине меридиана. Получается, что любая кривая на сфере может рассматриваться как сферический эллипс. По этой причине в определении эллипса должно быть внесено ограничение на выбор фокусов – они не должны быть концами одного диаметра. Риман именно на этом основании модифицировал сферическое пространство, отождествив диаметрально противоположные точки. Итак, в дальнейшем предполагается, что фокусы аналогов кривых второго порядка на сфере не являются диаметрально противоположными точками на сфере.

**Задание 1.** Доказать, что **любой сферический эллипс с фокусами  $F_1$  и  $F_2$  одновременно является сферической гиперболой с фокусами  $P_1$  и  $F_2$ , где  $P_1$  является точкой, диаметрально противоположной фокусу  $F_1$ .**

*Решение.* Пусть  $M$  точка на эллипсе, удовлетворяющая равенству  $F_1M + MF_2 = 2a$ . Расстояние между точками  $F_1$  и  $P_1$  равно длине меридиана  $m$ . Таким образом,  $F_1M + MP_1 = m$ . Из этого следует, что  $F_1M = m - MP_1$ . После подстановки этого выражения в исходное равенство получаем, что  $m - MP_1 + MF_2 = 2a$ . В итоге получаем уравнение гиперболы:  $MF_2 - MP_1 = 2a - m$ .

**Задание 2.** Доказать, что **любая сферическая гипербола одновременно является сферическим эллипсом.**

**Задание 3.** Доказать, что **любая сферическая парабола одновременно является сферическим эллипсом.**

*Решение.* Пусть  $M$  точка на сферической параболе, равноудалённая от фокуса  $F$  и от сферической прямой с полюсом  $P$ , лежащим на одной полусфере с точкой  $M$ . Проведём через точку  $M$  и диаметр сферы с концом  $P$  плоскость, которая пересекает директрису в точке  $A$ . Тогда уравнение, задающее сферическую параболу, имеет вид:  $AM = MF$ . Расстояние  $AP$  равно половине длины меридиана  $\frac{m}{2}$ . Значит,  $AM + MP = \frac{m}{2}$ . Отсюда следует, что  $MF + MP = \frac{m}{2}$ .

**Задание 4.** Описать все **сферические параболы**.

*Решение.* В предыдущем задании, отправляясь от общего определения сферической параболы, мы получили уравнение эллипса, которым является данная



парабола. При этом сумма расстояний до фокусов, оказалась равной половине длины меридиана. Таким образом, было получено необходимое условие для того, чтобы сферический эллипс являлся одновременно параболой. Теперь нужно проверить достаточность этого условия. Итак, пусть задан сферический эллипс с фокусами  $F_1$  и  $F_2$ , описываемый условием  $F_1M + MF_2 = \frac{m}{2}$ . Рассмотрим точку  $P$ , диаметрально противоположную фокусу  $F_2$  и линию большого круга, лежащую в плоскости, перпендикулярной диаметру  $PF_2$ . Эта линия большого круга должна оказаться директрисой. Проверим это.

Проведём через точку  $M$  и диаметр  $PF_2$  плоскость, которая пересекает предполагаемую директрису в точке  $A$ . Тогда  $AM + MF_2 = \frac{m}{2}$ . Таким образом,  $MF_2 = \frac{m}{2} - AM$ . Из этого следует, что  $F_1M + \frac{m}{2} - AM = \frac{m}{2}$ , а в итоге получаем, что  $F_1M = AM$ . Это и есть условие, задающее сферическую параболу.

Отметим, что вместо фокуса  $F_2$  можно было с тем же успехом рассмотреть и фокус  $F_1$ . Из этого следует, что у сферической параболы две директрисы.

**Задание 5.** Доказать, что **любая линия большого круга на сфере одновременно является прямой, окружностью и эллипсом.**

*Указание.* Для доказательства того, что линия большого круга является эллипсом, нужно провести плоскость через сопряжённый диаметр  $AB$ . Полученное таким образом сечение сферы очерчено двумя меридианами. На каждом из них выберем по точке  $F_1$  и  $F_2$ . При этом необходимо соблюсти условие  $AF_1 = AF_2$ . Теперь нужно доказать, что точки  $F_1$  и  $F_2$  являются фокусами линия большого круга.

Для рассмотренных выше сферических аналогов кривых второго порядка можно разработать сходные с методом Роберваля приёмы построения касательных к сферическим эллипсам. И на сфере этот подход основан на кинематических соображениях и состоит в том, что касательная и нормаль к эллипсу являются биссектрисами углов между фокальными радиусами. Таким образом, касательные к сферическим эллипсам являются биссектрисами углов между фокальными радиусами.

**Задание 6.** Доказать, что если на сфере заданы две точки  $F_1$  и  $F_2$ , а точка  $F_3$  является диаметрально противоположной к точке  $F_1$ , то любой сферический эллипс с фокусами  $F_1$  и  $F_2$  ортогонален любому сферическому эллипсу с фокусами  $F_2$  и  $F_3$ .

*Указание.* Нужно использовать результаты задания 1.

Следствием последнего задания является тот факт, что семейства эллипсов с фокусами  $F_1$  и  $F_2$  и эллипсов с фокусами  $F_2$  и  $F_3$  взаимно ортогональны. Это позволяет строить на сфере эффектные семейства кривых. К вопросу о построении изображений аналогов кривых второго порядка на сфере мы и перейдём.

## 4. ПОСТРОЕНИЕ КРИВЫХ НА СФЕРЕ

Если до сих пор удавалось обходиться без сколь-нибудь сложных вычислений, то далее будут производиться достаточно пространственные вычисления. Математической основой этих вычислений будет следующий круг понятий.



- Формулы аффинных преобразований трёхмерного пространства.
- Аксонометрия, как средство построения на плоскости объёмных изображений.
- Сферическая система координат.
- Скалярное произведение векторов, которое позволяет, отправляясь от сферических координат двух точек, вычислить расстояние между ними, выражаемое в угловой мере.
- Векторное произведение векторов, которое позволяет строить ортонормированные базисы в пространстве.

**Задание 7.** Построить на экране изображение сферы с прочерченной сетью параллелей и меридианов.

*Решение.* Далее приведена программа построения сферы. Результаты её работы показаны на рисунке 1.

```
pi = Math.Pi
GraphicsWindow.Width = 700
GraphicsWindow.Height = 700
'Координаты центра сферы, радиус и единичный отрезок
x0 = 350
y0 = 350
ed = 100
r = 2
'Базис аксонометрической системы координат
k = .25
u0 = pi/6
x1 = x0 + r*ed* Math.Cos(u0)
y1 = y0 - r*ed*k* Math.Sin(u0)
x2 = x0 + r*ed* Math.Cos(u0+pi/2)
y2 = y0 - r*ed*k* Math.Sin(u0+pi/2)
x3 = x0
y3 = y0 - r*ed
'Построение параллелей на сфере
For u1 = 0 To 2*pi Step pi/1000
For u2 = -pi/2 To pi/2 Step pi/10
p = Math.Cos(u1)*Math.Cos(u2)
q = Math.Sin(u1)*Math.Cos(u2)
w = Math.Sin(u2)
'Аффинные преобразования
x = x0 + p*(x1 - x0) + q*(x2 - x0) + w*(x3 - x0)
y = y0 + p*(y1 - y0) + q*(y2 - y0) + w*(y3 - y0)
GraphicsWindow.SetPixel(x, y,"")
EndFor
EndFor
'Построение меридианов на сфере
For u1 = 0 To 2*pi Step pi/10
For u2 = -pi/2 To pi/2 Step pi/1000
p = Math.Cos(u1)*Math.Cos(u2)
```



```
q = Math.Sin(u1)*Math.Cos(u2)
w = Math.Sin(u2)
x = x0 + p*(x1 - x0) + q*(x2 - x0) + w*(x3 - x0)
y = y0 + p*(y1 - y0) + q*(y2 - y0) + w*(y3 - y0)
GraphicsWindow.SetPixel(x, y, "")
EndFor
EndFor
```




Рис. 1

**Задание 8.** Выделить на сфере область, ограниченную сферическим эллипсом.

*Решение.* Используем следующий способ. После построения объёмного изображения сферы следует задать сферические координаты фокусов эллипса и вычислить координаты векторов, соединяющих центр сферы с этими фокусами. Далее начинается сканирование точек сферы. Вычисляются косинусы углов, которые образуют фокальные вектора с вектором, соединяющим текущую точку с центром сферы. Наконец вычисляется косинус суммы этих углов, и производится проверка того, превосходит ли он косинус заданного суммарного угла, определяющего эллипс. Если он больше, то текущая точка является внутренней точкой эллиптической области. Её следует отметить на сфере.

Приведённые ниже строки программы следует добавить к программе из предыдущего задания. Результат работы показан на рисунке 2.

```
'Первый фокус эллипса
fpu1 = pi + pi/3
fmu1 = 0
pf1 = Math.Cos(fpu1)*Math.Cos(fmu1)
qf1 = Math.Sin(fpu1)*Math.Cos(fmu1)
wfl = Math.Sin(fmu1)
x = x0 + pf1*(x1 - x0) + qf1*(x2 - x0) + wfl*(x3 - x0)
y = y0 + pf1*(y1 - y0) + qf1*(y2 - y0) + wfl*(y3 - y0)
GraphicsWindow.FillEllipse(x-5, y-5,10,10)
```



*'Второй фокус эллипса*

fpu2 = fpu1 + 2\*pi/7

fmu2 = 0

pf2 = Math.Cos(fpu2)\*Math.Cos(fmu2)

qf2 = Math.Sin(fpu2)\*Math.Cos(fmu2)

wf2 = Math.Sin(fmu2)

x = x0 + pf2\*(x1 - x0) + qf2\*(x2 - x0) + wf2\*(x3 - x0)

y = y0 + pf2\*(y1 - y0) + qf2\*(y2 - y0) + wf2\*(y3 - y0)

GraphicsWindow.FillEllipse(x-5, y-5,10,10)

*'Сумма расстояний от фокусов до любой точки на эллипсе*

a = Math.Cos(2\*pi/5)

*'Сканирование сферы*

For u1 = pi To 2\*pi Step pi/400

For u2 = -pi/2 To pi/2 Step pi/400

*'Радиус, направленный в текущую точку*

p = Math.Cos(u1)\*Math.Cos(u2)

q = Math.Sin(u1)\*Math.Cos(u2)

w = Math.Sin(u2)

*'Скалярные произведения текущего и фокальных радиусов*

cs1 = pf1\*p+qf1\*q+wf1\*w

sn1 = Math.SquareRoot(1 - cs1\*cs1)

cs2 = pf2\*p+qf2\*q+wf2\*w

sn2 = Math.SquareRoot(1 - cs2\*cs2)

*'Косинус суммы углов между текущим и фокальными радиусами*

sum = cs1\*cs2 - sn1\*sn2

If sum > a Then

*'Аффинные преобразования*

x = x0 + p\*(x1 - x0) + q\*(x2 - x0) + w\*(x3 - x0)

y = y0 + p\*(y1 - y0) + q\*(y2 - y0) + w\*(y3 - y0)

GraphicsWindow.SetPixel(x, y,"red")

EndIf

EndFor

EndFor




Рис. 2





Рассмотренный выше способ визуализации кривых на сфере обладает рядом недостатков. По этой причине необходимо найти более эффективные пути построения подобных кривых. Начнём с построения линий большого круга.

**Задание 9.** Пусть на сфере задана точка  $P$ . Построить линию большого круга, для которой эта точка является полюсом.

*Решение.* Зададим полюс  $P_1$  сферическими координатами и вычислим координаты вектора  $OP_1 = V_1 = (p; q; w)$ , соединяющего центр сферы с этим полюсом. Искомая линия большого круга лежит в плоскости, перпендикулярной этому вектору. Уравнение этой плоскости имеет вид  $p \cdot x + q \cdot y + w \cdot z = 0$ . Если  $w = 0$ , то линия большого круга является одной из параллелей. Её построение не составляет труда. Если же  $w$  не равно нулю, то плоскость  $p \cdot x + q \cdot y + w \cdot z = 0$  пересекается с горизонтальной плоскостью по прямой  $p \cdot x + q \cdot y = 0$ . Вдоль этой прямой направлен единичный вектор  $OP_2 = V_2 = \left( -\frac{q}{\sqrt{p^2 + q^2}}; \frac{p}{\sqrt{p^2 + q^2}}; 0 \right)$ . Найдём векторное произведение векторов  $V_1$  и  $V_2$  и получим вектор  $OP_3 = V_3$ , лежащий в плоскости линии большого круга. Если в базисе  $V_2$  и  $V_3$  построить окружность с помощью параметрических уравнений, то это и будет искомая линия большого круга. Остаётся добавить, что

$$V_3 = \left( \begin{array}{c|c|c} \frac{q}{\sqrt{p^2 + q^2}} & w & \left| \begin{array}{c|c} p & w \\ \hline -\frac{q}{\sqrt{p^2 + q^2}} & 0 \end{array} \right| \\ \hline \frac{p}{\sqrt{p^2 + q^2}} & 0 & \left| \begin{array}{c|c} p & q \\ \hline -\frac{q}{\sqrt{p^2 + q^2}} & \frac{p}{\sqrt{p^2 + q^2}} \end{array} \right| \end{array} \right) = \left( -\frac{pw}{\sqrt{p^2 + q^2}}; -\frac{qw}{\sqrt{p^2 + q^2}}; \sqrt{p^2 + q^2} \right).$$

Приведённые ниже строки программы, как и ранее, следует добавить к программе из задания 7. Результат работы показан на рисунке 3.

*‘Центр сферы*

GraphicsWindow.BrushColor = «black»

GraphicsWindow.FillEllipse(x0-5, y0-5, 10, 10)

*‘Полюс и базисные вектора*

u1 = 3\*pi/2

u2 = pi/3

p1 = Math.Cos(u1)\*Math.Cos(u2)

q1 = Math.Sin(u1)\*Math.Cos(u2)

w1 = Math.Sin(u2)

x = x0 + p1\*(x1 - x0) + q1\*(x2 - x0) + w1\*(x3 - x0)

y = y0 + p1\*(y1 - y0) + q1\*(y2 - y0) + w1\*(y3 - y0)

GraphicsWindow.FillEllipse(x-5, y-5, 10, 10)

GraphicsWindow.DrawLine(x, y, x0, y0)

d = Math.Sqrt(p1\*p1 + q1\*q1)



```
p2 = -q1/d
q2 = p1/d
w2 = 0
x = x0 + p2*(x1 - x0) + q2*(x2 - x0) + w2*(x3 - x0)
y = y0 + p2*(y1 - y0) + q2*(y2 - y0) + w2*(y3 - y0)
GraphicsWindow.FillEllipse(x-5, y-5,10,10)
GraphicsWindow.DrawLine(x, y, x0, y0)
p3 = -p1*w1/d
q3 = -q1*w1/d
w3 = d
x = x0 + p3*(x1 - x0) + q3*(x2 - x0) + w3*(x3 - x0)
y = y0 + p3*(y1 - y0) + q3*(y2 - y0) + w3*(y3 - y0)
GraphicsWindow.FillEllipse(x-5, y-5,10,10)
GraphicsWindow.DrawLine(x, y, x0, y0)
'Линия большого круга
For ug = 0 To 2*pi Step pi/1000
t1 = p2 * Math.Cos(ug) + p3 * Math.Sin(ug)
t2 = q2 * Math.Cos(ug) + q3 * Math.Sin(ug)
t3 = w2 * Math.Cos(ug) + w3 * Math.Sin(ug)
x = x0 + t1*(x1 - x0) + t2*(x2 - x0) + t3*(x3 - x0)
y = y0 + t1*(y1 - y0) + t2*(y2 - y0) + t3*(y3 - y0)
GraphicsWindow.FillEllipse(x-1, y-1,2,2)
EndFor
```




Рис. 3

**Задание 10.** Построить линию большого круга, проходящую через две точки А и В на сфере.

*Указание.* Нужно в плоскости АОВ, где О – центр сферы, построить вектор ОС, перпендикулярный вектору АО. Затем в базисе АО и СО по аналогии с предыдущим заданием строится либо линия большого круга, либо дуга АВ.



Построенная в задании 9 система координат из трёх векторов позволяет именно к ней привязывать сферические координаты. Рассмотрение группы преобразований трёхмерного пространства с неподвижной точкой  $O$ , как раз и связана с переходом от одного базиса к другому. Параллельное использование сферических координат позволяет описать группу движений сферической геометрии. Легко понять, что эта тематика позволяет поставить перед студентом ряд интересных и содержательных заданий. Мы же ограничимся только одним заданием.

**Задание 11.** Используя базис из задания 9, построить привязанную к нему систему параллелей и меридианов на сфере.

*Решение.* Положение текущей точки на сфере определяется сферическими координатами и тем базисом, к которому привязаны сферические координаты. Итак, сначала вычисляются три сферические координаты  $(s_1; s_2; s_3)$ . Затем эти координаты рассматриваются в базисе  $V_2, V_2$  и  $V_3$  и переводятся в координаты аксонометрической системы  $(t_1; t_2; t_3)$ . Наконец, аксонометрические координаты преобразуются в экранные. Результат работы программы показан на рисунке 4.

```
pi = Math.Pi
GraphicsWindow.Width = 700
GraphicsWindow.Height = 700
'Координаты центра сферы, радиус и единичный отрезок
x0 = 350
y0 = 350
ed = 300
'Базис аксонометрической системы координат
k = .25
u0 = pi/6
x1 = x0 + ed* Math.Cos(u0)
y1 = y0 - ed*k* Math.Sin(u0)
x2 = x0 + ed* Math.Cos(u0+pi/2)
y2 = y0 - ed*k* Math.Sin(u0+pi/2)
x3 = x0
y3 = y0 - ed
'Центр сферы
GraphicsWindow.BrushColor = "black"
GraphicsWindow.FillEllipse(x0-5, y0-5,10,10)
'Полус P1 и базисные вектора
u1 = 3*pi/2
u2 = pi/3
p1 = Math.Cos(u1)*Math.Cos(u2)
q1 = Math.Sin(u1)*Math.Cos(u2)
w1 = Math.Sin(u2)
x = x0 + p1*(x1 - x0) + q1*(x2 - x0) + w1*(x3 - x0)
y = y0 + p1*(y1 - y0) + q1*(y2 - y0) + w1*(y3 - y0)
GraphicsWindow.FillEllipse(x-5, y-5,10,10)
GraphicsWindow.DrawLine(x, y, x0, y0)
```



```
d = Math.SquareRoot(p1*p1 + q1*q1)
p2 = -q1/d
q2 = p1/d
w2 = 0
x = x0 + p2*(x1 - x0) + q2*(x2 - x0) + w2*(x3 - x0)
y = y0 + p2*(y1 - y0) + q2*(y2 - y0) + w2*(y3 - y0)
GraphicsWindow.FillEllipse(x-5, y-5, 10, 10)
GraphicsWindow.DrawLine(x, y, x0, y0)
p3 = -p1*w1/d
q3 = -q1*w1/d
w3 = d
x = x0 + p3*(x1 - x0) + q3*(x2 - x0) + w3*(x3 - x0)
y = y0 + p3*(y1 - y0) + q3*(y2 - y0) + w3*(y3 - y0)
GraphicsWindow.FillEllipse(x-5, y-5, 10, 10)
GraphicsWindow.DrawLine(x, y, x0, y0)
'Построение параллелей на сфере
For u1 = 0 To 2*pi Step pi/1000
For u2 = -pi/2 To pi/2 Step pi/10
'Сферические координаты
s1 = Math.Cos(u1)*Math.Cos(u2)
s2 = Math.Sin(u1)*Math.Cos(u2)
s3 = Math.Sin(u2)
'Координаты в наклонной системе координат
t1 = p1 * s1 + p2 * s2 + p3 * s3
t2 = q1 * s1 + q2 * s2 + q3 * s3
t3 = w1 * s1 + w2 * s2 + w3 * s3
'Экранные координаты
x = x0 + t1*(x1 - x0) + t2*(x2 - x0) + t3*(x3 - x0)
y = y0 + t1*(y1 - y0) + t2*(y2 - y0) + t3*(y3 - y0)
GraphicsWindow.SetPixel(x, y, "")
EndFor
EndFor
'Построение меридианов на сфере
For u1 = 0 To 2*pi Step pi/10
For u2 = -pi/2 To pi/2 Step pi/1000
'Сферические координаты
s1 = Math.Cos(u1)*Math.Cos(u2)
s2 = Math.Sin(u1)*Math.Cos(u2)
s3 = Math.Sin(u2)
'Координаты в наклонной системе координат
t1 = p1 * s1 + p2 * s2 + p3 * s3
t2 = q1 * s1 + q2 * s2 + q3 * s3
t3 = w1 * s1 + w2 * s2 + w3 * s3
'Экранные координаты
x = x0 + t1*(x1 - x0) + t2*(x2 - x0) + t3*(x3 - x0)
y = y0 + t1*(y1 - y0) + t2*(y2 - y0) + t3*(y3 - y0)
GraphicsWindow.SetPixel(x, y, "")
EndFor
EndFor
```




Рис. 4

Перейдём к вопросу о достаточно эффективном способе построения сферических эллипсов. Немецкий математик, занимавшийся проективной геометрией, Людвиг Иммануил Магнус показал, что **сферический эллипс является линией пересечения шара с конусом второго порядка, вершина которого лежит в центре шара**. Именно это обстоятельство можно использовать при построениях. Однако начнём с доказательства приведённого выше утверждения.

Построения будем проводить, отправляясь от сферы единичного радиуса с центром в точке  $H$ . Расстояние между двумя точками  $P$  и  $Q$  будет задаваться величиной центрального угла  $\angle PHQ$ .

Пусть сферический эллипс определяется следующими параметрами (рисунок 5). Расстояние от центра  $O$  до каждого из фокусов определяется углом  $\varphi$ . Полуось сферического эллипса определяется углом  $\alpha$ . С помощью центральной проекции отобразим сферический эллипс на горизонтальную плоскость. При этом на данный момент можно говорить только о положении некоторых точек и длине некоторых отрезков.

Перечислим эти точки. Отрезок  $OH$  перпендикулярен горизонтальной плоскости и имеет единичную длину. Полуось сферического эллипса отображается в отрезок  $OA$ . Угол  $\angle AHO$  равен  $\alpha$ . Из этого следует, что полуось  $OA = a$  плоского эллипса, в который был отображён эллипс сферический, равна  $tg \alpha$ . Фокусы сферического эллипса отображаются в точки  $F_1$  и  $F_2$ . Углы  $\angle F_1HO$  и  $\angle F_2HO$  равны  $\varphi$ . Расстояние от центра плоского эллипса до каждой из проекций  $F_1$  и  $F_2$  фокусов сферического эллипса равно  $tg \varphi$ . Что касается длины проекции  $OB$  на плоскость второй полуоси сферического эллипса, то её нужно вычислять.

Пусть угол  $\angle OHB$  равен  $\beta$  и  $OB = b = tg \beta$ . Выразим тангенс  $\beta$  через значения  $\alpha$  и  $\varphi$ . Возникающее по первому впечатлению предположение о том, что точки  $F_1$  и  $F_2$  являются фокусами плоского эллипса, является неверным. Дело в том, что речь идёт не о постоянстве суммы двух длин, а о постоянстве суммы двух углов. В частности  $\angle AHF_1 + \angle AHF_2 = 2\alpha$  и  $\angle BHF_1 + \angle BHF_2 = 2\alpha$ . Именно этот факт позволяет вычислить  $tg \beta$ .




Рис. 5

По теореме Пифагора  $F_1H = \frac{1}{\cos \varphi}$ ,  $HB = \frac{1}{\cos \beta}$  и  $F_1B^2 = tg^2 \varphi + tg^2 \beta$ . Применим теорему косинусов к треугольнику  $BHF_1$  и в итоге получим, что  $\cos \beta = \frac{\cos \alpha}{\cos \varphi}$ . Далее получаем, что  $\sin \beta = \frac{\sqrt{\cos^2 \varphi - \cos^2 \alpha}}{\cos \varphi}$  и, наконец,  $tg \beta = \frac{\sqrt{\cos^2 \varphi - \cos^2 \alpha}}{\cos \alpha}$ .

Теперь мы можем предположить, что при перемещении точки  $M$  по плоскому эллипсу с полуосями  $a = tg \alpha$  и  $b = tg \beta$ , сумма углов  $u_1 = \angle MHF_1$  и  $u_2 = \angle MHF_2$  постоянна и равна  $2\alpha$  (рисунок 6). Именно этот факт и необходимо проверить.




Рис. 6

Приступим к выражению этого утверждения в виде формулы. Введём обозначения для основных углов:  $\angle MHF_1 = u_1$  и  $\angle MHF_2 = u_2$ . Нужно доказать, что  $u_1 + u_2 = 2\alpha$ . Поскольку вычисление углов более затруднено, чем вычисление их тригонометрических функций, будем доказывать, что  $\cos(u_1 + u_2) = \cos 2\alpha$ .

Чтобы вычислить  $\cos u_1$  и  $\cos u_2$  используем скалярные произведения векторов  $\overline{HF_1} \cdot \overline{HM}$  и  $\overline{HF_2} \cdot \overline{HM}$ . Прежде всего,  $\overline{HF_1} = (-tg\varphi; 0; 1)$  и  $\overline{HF_2} = (tg\varphi; 0; 1)$ .



По нашему предположению точка  $M$  лежит на плоском эллипсе с полуосями  $a = tg \alpha$  и  $b = tg \beta$ . Таким образом, её координаты на плоскости можно задать параметрическими уравнениями  $x = a \cdot \cos u$  и  $y = b \cdot \sin u$ . Значит,  $\overline{HM} = (a \cdot \cos u; b \cdot \sin u; 1)$ . Теперь можно выразить  $\cos u_1$  и  $\cos u_2$  через тригонометрические функции углов  $\alpha$  и  $\varphi$ . Опуская достаточно простую цепочку преобразований, дадим окончательные выражения этих косинусов:

$$\cos u_1 = \frac{\cos \varphi \cdot \cos \alpha - \sin \varphi \cdot \sin \alpha \cdot \cos u}{\sqrt{\cos^2 u + \cos^2 \varphi \cdot \sin^2 u}}, \quad \cos u_2 = \frac{\cos \varphi \cdot \cos \alpha + \sin \varphi \cdot \sin \alpha \cdot \cos u}{\sqrt{\cos^2 u + \cos^2 \varphi \cdot \sin^2 u}}.$$

Эти формулы заставляют сделать предположение, что доказательство формулы  $\cos(u_1 + u_2) = \cos 2\alpha$  будет достаточно сложным. При этом пока нет полной уверенности, что исходная гипотеза верна. По этой причине её желательно проверить с помощью предварительных вычислений.

**Задание 12.** Написать программу, которая позволяет проверить истинность формулы  $\cos(u_1 + u_2) = \cos 2\alpha$  для конкретных значений  $\alpha$  и  $\varphi$ .

*Решение.* Результат работы программы показан на рисунке 7. Он показывает, что предположение верно.

```

pi = Math.Pi
r = 100
'Полуось a
al = pi/5
ca = Math.Cos(al)
ta = Math.Sqrt(1 - ca*ca)/ca
a = r*ta
'Параметр c
fi = pi/7
cf = Math.Cos(fi)
tf = Math.Sqrt(1 - cf*cf)/cf
c = r*tf
'Полуось b
cb = ca/cf
tb = Math.Sqrt(1 - cb*cb)/cb
b = r*tb
'Длина векторов FH
mv = Math.Sqrt(c*c + r*r)
i = 20
c2al = Math.Cos(2*al)
'Изменение параметра u
For u = 0 To 2*pi step pi/8
'Вектор MH
x = a* Math.Cos(u)
y = b* Math.Sin(u)
mt = Math.Sqrt(x*x + y*y + r*r)

```



```
'Вычисление косинуса и синуса угла u1  
c1 = (r*r - c*x)/(mv*mt)  
s1 = Math.Sqrt(1 - c1*c1)  
'Вычисление косинуса и синуса угла u2  
c2 = (r*r + c*x)/(mv*mt)  
s2 = Math.Sqrt(1 - c2*c2)  
'Вычисление косинуса угла u1 + u2  
csum = c1*c2 - s1*s2  
'Вычисление разности косинуса угла u1 + u2 и косинуса угла 2*a1  
GraphicsWindow.DrawText(50, i, csum - c2a1)  
i = i + 20  
EndFor
```




Рис. 7

Теперь можно перейти к доказательству формулы  $\cos(u_1 + u_2) = \cos 2\alpha$ . Но сначала отметим, что проверочное вычисление было полезным, поскольку вселило в нас уверенность в правильности исходной гипотезы. Если у приверженцев чисто теоретических методов работы с математическими проблемами возникнет пренебрежительное отношение к прагматической вычислительной проверке некоторых фактов, напомним им о методах Архимеда. Он не стеснялся вырезать из фольги геометрические фигуры и взвешивать их. Сейчас для сходных целей можно использовать компьютер.

Проведём необходимые для доказательства преобразования, опуская некоторые детали. Отправляясь от формулы  $\cos u_1 = \frac{\cos \varphi \cdot \cos \alpha - \sin \varphi \cdot \sin \alpha \cdot \cos u}{\sqrt{\cos^2 u + \cos^2 \varphi \cdot \sin^2 u}}$ , преобразуем выражение под корнем в знаменателе:





$$\begin{aligned} & \cos^2 u + \cos^2 \varphi \cdot \sin^2 u = \\ & = \cos^2 u + \cos^2 \varphi \cdot \sin^2 u + \sin^2 \varphi \cdot \sin^2 u - \sin^2 \varphi \cdot \sin^2 u = \\ & = \cos^2 u + \sin^2 u - \sin^2 \varphi \cdot \sin^2 u = 1 - \sin^2 \varphi \cdot \sin^2 u. \end{aligned}$$

Введём краткие обозначения для тригонометрических выражений, входящих в формулы для  $\cos u_1$  и  $\cos u_2$ :

$$p = \cos \varphi \cdot \cos \alpha - \sin \varphi \cdot \sin \alpha \cdot \cos u,$$

$$q = \cos \varphi \cdot \cos \alpha + \sin \varphi \cdot \sin \alpha \cdot \cos u,$$

$$r = 1 - \sin^2 \varphi \cdot \sin^2 u.$$

В этих обозначениях  $\cos u_1 = \frac{p}{\sqrt{r}}$  и  $\cos u_2 = \frac{q}{\sqrt{r}}$ . Далее  $\sin u_1 = \frac{\sqrt{r-p^2}}{\sqrt{r}}$  и  $\sin u_2 = \frac{\sqrt{r-q^2}}{\sqrt{r}}$ . Отсюда следует, что  $\cos(u_1 + u_2) = \frac{pq}{r} - \frac{\sqrt{(r-p^2)(r-q^2)}}{r}$ . Доказать нужно формулу  $\frac{pq}{r} - \frac{\sqrt{(r-p^2)(r-q^2)}}{r} = \cos 2\alpha$ .

Избавимся от радикалов и дробей и получим уравнение

$$(r-p^2)(r-q^2) = (pq - r \cos 2\alpha)^2.$$

После преобразований перейдём к уравнению

$$r \sin^2 2\alpha + 2pq \cos 2\alpha - p^2 - q^2 = 0.$$

Прибавив и отняв от левой части уравнения выражения член  $2pq$ , получим уравнение  $r \sin^2 2\alpha - (p+q)^2 + 2pq(1 + \cos 2\alpha) = 0$ . Теперь можно подставить исходные значения переменных  $p$ ,  $q$  и  $r$ :

$$\begin{aligned} & (1 - \sin^2 \varphi \cdot \sin^2 u) \cdot \sin^2 2\alpha - 4 \cdot \cos^2 \varphi \cdot \cos^2 \alpha + \\ & + 4 \cdot (\cos^2 \varphi \cdot \cos^2 \alpha - \sin^2 \varphi \cdot \sin^2 \alpha \cdot \cos^2 u) = 0 \end{aligned}$$

Доказательство этого тождества проводится без труда. Но отметим, что проверку правильности проведённых к определённому моменту преобразований с помощью вычисления выражений при конкретных значениях переменных можно было проводить несколько раз.

Итак, теорема Магнуса доказана. Более того, показано, что фокусы сферического и плоского эллипса друг в друга не проецируются. Наконец, получены формулы, позволяющие построить плоский эллипс, связанный с эллипсом сферическим.

Перейдём к вопросам, связанным с эффективным методом построения сферического эллипса.

**Задание 13.** Разработать алгоритм построения сферического эллипса в аксонометрической проекции.

*Решение.* Исходными данными для построения сферического эллипса являются:



- Сферические координаты фокусов эллипса.
- Угловая мера  $\alpha$  полуоси эллипса.

Построение будет начато с получения изображения сферы с центром Н (смотри задание 7). Далее необходимо провести ряд вычислений.

1. Вычислить координаты фокусов сферического эллипса  $\Phi_1$  и  $\Phi_2$  в трёхмерном пространстве и отобразить их на экране.
2. Вычислить координаты центра сферического эллипса О и отобразить её на экране.
3. Вычислить координаты вектора НО.
4. Получить уравнение плоскости, проходящей через точку О перпендикулярно вектору НО.
5. Определить координаты первого базисного вектора ОР на этой плоскости, нормируя вектор  $\Phi_1\Phi_2$ .
6. Определить координаты второго базисного вектора ОТ на этой плоскости, как векторное произведение ОН и ОР.
7. Если нужно, построить изображение плоскости РОТ.
8. Определить полуоси плоского эллипса на плоскости РОТ.
9. С помощью параметрического уравнения вычислить координаты текущей точки М на плоском эллипсе.
10. Нормируя вектор НМ построить соответствующую точку  $M_1$  сферического эллипса.

**Задание 14.** Написать программу построения сферического эллипса.

*Решение.* Результат работы программы показан на рисунке 8.

```
pi = Math.Pi
GraphicsWindow.Width = 700
GraphicsWindow.Height = 700
'Координаты центра сферы, радиус и единичный отрезок
x0 = 350
y0 = 350
ed = 200
'Базис аксонометрической системы координат
k = .25
u0 = pi/6
x1 = x0 + ed* Math.Cos(u0)
y1 = y0 - ed*k* Math.Sin(u0)
x2 = x0 + ed* Math.Cos(u0+pi/2)
y2 = y0 - ed*k* Math.Sin(u0+pi/2)
x3 = x0
y3 = y0 - ed
'Построение параллелей на сфере
For u1 = 0 To 2*pi Step pi/1000
For u2 = -pi/2 To pi/2 Step pi/10
p = Math.Cos(u1)*Math.Cos(u2)
q = Math.Sin(u1)*Math.Cos(u2)
```



```
w = Math.Sin(u2)
'Аффинные преобразования
x = x0 + p*(x1 - x0) + q*(x2 - x0) + w*(x3 - x0)
y = y0 + p*(y1 - y0) + q*(y2 - y0) + w*(y3 - y0)
GraphicsWindow.SetPixel(x, y,"")
EndFor
EndFor
'Построение меридианов на сфере
For u1 = 0 To 2*pi Step pi/10
For u2 = -pi/2 To pi/2 Step pi/1000
p = Math.Cos(u1)*Math.Cos(u2)
q = Math.Sin(u1)*Math.Cos(u2)
w = Math.Sin(u2)
'Аффинные преобразования
x = x0 + p*(x1 - x0) + q*(x2 - x0) + w*(x3 - x0)
y = y0 + p*(y1 - y0) + q*(y2 - y0) + w*(y3 - y0)
GraphicsWindow.SetPixel(x, y,"")
EndFor
EndFor
'Центр сферы
GraphicsWindow.BrushColor = "black"
GraphicsWindow.FillEllipse(x0-5, y0-5,10,10)
'Первый фокус
u11 = .6*pi
u21 = 0
p1 = Math.Cos(u11)*Math.Cos(u21)
q1 = Math.Sin(u11)*Math.Cos(u21)
w1 = Math.Sin(u21)
x = x0 + p1*(x1 - x0) + q1*(x2 - x0) + w1*(x3 - x0)
y = y0 + p1*(y1 - y0) + q1*(y2 - y0) + w1*(y3 - y0)
GraphicsWindow.FillEllipse(x-5, y-5,10,10)
GraphicsWindow.DrawLine(x, y, x0, y0)
'Второй фокус
u12 = 1*pi
u22 = .2
p2 = Math.Cos(u12)*Math.Cos(u22)
q2 = Math.Sin(u12)*Math.Cos(u22)
w2 = Math.Sin(u22)
x = x0 + p2*(x1 - x0) + q2*(x2 - x0) + w2*(x3 - x0)
y = y0 + p2*(y1 - y0) + q2*(y2 - y0) + w2*(y3 - y0)
GraphicsWindow.FillEllipse(x-5, y-5,10,10)
GraphicsWindow.DrawLine(x, y, x0, y0)
'Центр эллипса
p0 = p1 + p2
q0 = q1 + q2
w0 = w1 + w2
d = Math.Sqrt(p0*p0 + q0*q0 + w0*w0)
```



```
p0 = p0/d
q0 = q0/d
w0 = w0/d
x = x0 + p0*(x1 - x0) + q0*(x2 - x0) + w0*(x3 - x0)
y = y0 + p0*(y1 - y0) + q0*(y2 - y0) + w0*(y3 - y0)
GraphicsWindow.FillEllipse(x-5, y-5,10,10)
GraphicsWindow.DrawLine(x, y, x0, y0)
'Первый базисный вектор на плоскости
pb1 = p2 - p1
qb1 = q2 - q1
wb1 = w2 - w1
d = Math.Sqrt(pb1*pb1 + qb1*qb1 + wb1*wb1)
pb1 = pb1/d
qb1 = qb1/d
wb1 = wb1/d
x = x0 + p0*(x1 - x0) + q0*(x2 - x0) + w0*(x3 - x0)
y = y0 + p0*(y1 - y0) + q0*(y2 - y0) + w0*(y3 - y0)
GraphicsWindow.FillEllipse(x-5, y-5,10,10)
GraphicsWindow.DrawLine(x, y, x0, y0)
'Второй базисный вектор на плоскости
pb2 = q0*wb1 - qb1*w0
qb2 = -(p0*wb1 - pb1*w0)
wb2 = p0*qb1 - pb1*q0
x = x0 + pb2*(x1 - x0) + qb2*(x2 - x0) + wb2*(x3 - x0)
y = y0 + pb2*(y1 - y0) + qb2*(y2 - y0) + wb2*(y3 - y0)
GraphicsWindow.FillEllipse(x-5, y-5,10,10)
GraphicsWindow.DrawLine(x, y, x0, y0)
'Построение плоскости
For t1 = -2 To 2 Step .01
For t2 = -1 To 1 Step .1
tt1 = p0 + pb1*t1 + pb2*t2
tt2 = q0 + qb1*t1 + qb2*t2
tt3 = w0 + wb1*t1 + wb2*t2
x = x0 + tt1*(x1 - x0) + tt2*(x2 - x0) + tt3*(x3 - x0)
y = y0 + tt1*(y1 - y0) + tt2*(y2 - y0) + tt3*(y3 - y0)
GraphicsWindow.SetPixel(x, y,"")
EndFor
EndFor
For t1 = -2 To 2 Step .25
For t2 = -1 To 1 Step .01
tt1 = p0 + pb1*t1 + pb2*t2
tt2 = q0 + qb1*t1 + qb2*t2
tt3 = w0 + wb1*t1 + wb2*t2
x = x0 + tt1*(x1 - x0) + tt2*(x2 - x0) + tt3*(x3 - x0)
y = y0 + tt1*(y1 - y0) + tt2*(y2 - y0) + tt3*(y3 - y0)
GraphicsWindow.SetPixel(x, y,"")
EndFor
```



```
EndFor
```

```
  'Построение плоского и сферического эллипсов
```

```
cf = p0*p1 + q0*q1 + w0*w1
```

```
fu = Math.ArcCos(cf)
```

```
al = fu + .1
```

```
ca = Math.Cos(al)
```

```
a = Math.Tan(al)
```

```
b = Math.Sqrt(cf*cf - ca*ca)/ca
```

```
For ug = 0 To 2*pi Step pi/1000
```

```
  'Точка на плоском эллипсе
```

```
t1 = a * Math.Cos(ug)
```

```
t2 = b * Math.Sin(ug)
```

```
tt1 = p0 + pb1*t1 + pb2*t2
```

```
tt2 = q0 + qb1*t1 + qb2*t2
```

```
tt3 = w0 + wb1*t1 + wb2*t2
```

```
x = x0 + tt1*(x1 - x0) + tt2*(x2 - x0) + tt3*(x3 - x0)
```

```
y = y0 + tt1*(y1 - y0) + tt2*(y2 - y0) + tt3*(y3 - y0)
```

```
GraphicsWindow.FillEllipse(x-1, y-1,2,2)
```

```
  'Точка на сферическом эллипсе
```

```
d = Math.Sqrt(tt1*tt1 + tt2*tt2 + tt3*tt3)
```

```
tt1 = tt1/d
```

```
tt2 = tt2/d
```

```
tt3 = tt3/d
```

```
x = x0 + tt1*(x1 - x0) + tt2*(x2 - x0) + tt3*(x3 - x0)
```

```
y = y0 + tt1*(y1 - y0) + tt2*(y2 - y0) + tt3*(y3 - y0)
```

```
GraphicsWindow.FillEllipse(x-2, y-2,4,4)
```

```
EndFor
```




Рис. 8



### **Темы для самостоятельной работы студентов, связанные со сферикой.**

1. На сфере заданы три точки  $A$ ,  $B$  и  $C$ . Построить сферический треугольник  $ABC$ .
2. Построить описанную окружность сферического треугольника  $ABC$ .
3. Построить вписанную окружность сферического треугольника  $ABC$ .
4. Построить на сфере аналог спирали Архимеда.
5. Построить изображения икосаэдра и додекаэдра.
6. Построить на сфере семейство софокусных эллипсов.
7. Построить на сфере семейство софокусных эллипсов и семейство ортогональных к ним гипербол.
8. Разработать программу управления аналогом черепашки из языка Лого на сфере.

### **Темы для самостоятельной работы студентов, связанные с построением объёмных изображений.**

1. Разработать программу построения изображения плоскости по трём точкам в пространстве.
2. Написать программы для построения поверхностей, описываемых функциями от двух переменных.
3. Построить изображение тора.
4. Построить изображение листа Мёбиуса.
5. Построить изображение бутылки Клейна.
6. Разработать методы построения различных поверхностей второго порядка.
7. Написать программы, создающие анимационные эффекты с объёмными изображениями, например, вращение пирамид и призм.

## **5. ЗАКЛЮЧЕНИЕ**

Что касается вопросов, связанных с вычислениями в различных областях геометрии, то здесь открывается широчайший круг для методических разработок. Тем не менее, авторы надеются, что и методические идеи, изложенные в данной статье и связанные с достаточно специфичной проблематикой, могут пригодиться молодым преподавателям и помочь в их нелёгкой работе. Предполагается написание ещё ряда статей, связанных с вычислениями в различных областях математики.

### ***Литература***

1. Куланин Е.Д., Нуркаева И.М. О двух геометрических задачах на экстремум. Математика в школе. 2019. № 4. С. 35–40.
2. Куланин Е.Д., Нуркаева И.М. Еще раз о задаче Мавло. Математика в школе. 2020. № 2. С. 76–79.
3. Куланин Е.Д., Степанов М.Е., Нуркаева И.М. Пропедевтика решения экстремальных задач в школьном курсе математики. Моделирование и анализ данных. 2019. № 4. С. 127–144.
4. Куланин Е.Д., Нгуен Ву Куанг, Степанов М.Е. Осязаемая предметность с компьютерной поддержкой. Моделирование и анализ данных. Научный журнал. 2019. № 4. С. 145–156.
5. Куланин Е.Д., Степанов М.Е., Нуркаева И.М. Роль образного мышления в научном мышлении. Моделирование и анализ данных. 2020. Т. 10. № 2 С. 110–128.



6. Куланин Е.Д., Степанов М.Е., Нуркаева И.М. О различных подходах к решению экстремальных задач. Моделирование и анализ данных. 2020. Т. 11. № 1. С. 40–60.
7. Лунгу К.Н., Норин В.П., Письменный Д.Т., Шевченко Ю.А., Куланин Е.Д. Сборник задач по высшей математике с контрольными работами. Москва, 2013. Том 2 (8-е издание).
8. Степанов М.Е. Некоторые вопросы методики преподавания высшей математики. Моделирование и анализ данных. 2017. № 1. С. 54–94.
9. Куланин Е.Д., Степанов М.Е. Из опыта работы в режиме дистанционного обучения Моделирование и анализ данных. 2022. Т. 12. № 3. С. 58–70.
10. Куланин Е.Д., Степанов М.Е. Всестороннее рассмотрение математических понятий как методический прием. Моделирование и анализ данных. 2022. Т. 12. № 4. С. 67–84.
11. Куланин Е.Д., Степанов М.Е. О визуализации решений некоторых экстремальных задач. Моделирование и анализ данных. 2022. Т. 12. № 4. С. 94–104.
12. Куланин Е.Д., Степанов М.Е., Панфилов А.Д., Потоньшев И.С. Системный подход к методике тифлопедагогики на примере задач математического анализа. 2022. Т. 12. № 2. С. 34–82.
13. Куланин Е.Д., Степанов М.Е. Вычислительный эксперимент в преподавании высшей математики на примере теории чисел. Моделирование и анализ данных. 2024. Т. 14. № 1. С. 170–195.
14. Энциклопедия элементарной математики. Книга четвёртая. Геометрия. М., Гос. изд. физ.-мат. лит., 1963.
15. Лысенко В.И. Николай Иванович Фусс, М., Наука, 1975.



## Computational Experiment in Teaching Higher Mathematics. Geometry

***Yevgeny D. Kulanin\****

Moscow State University of Psychology & Education (MSUPE), Moscow, Russia

ORCID: <https://orcid.org/0000-0001-6093-7012>

e-mail: [lucas03@mail.ru](mailto:lucas03@mail.ru)

***Mikhail E. Stepanov\*\****

Moscow State University of Psychology & Education (MSUPE), Moscow, Russia

ORCID: <https://orcid.org/0000-0003-4803-8211>

e-mail: [mestepanov@yandex.ru](mailto:mestepanov@yandex.ru)

The article continues the cycle ([1] – [13]) of methodological developments of the authors. It discusses some problems related to ways to improve the culture of mathematical thinking of mathematics students. The authors rely on their work experience at the Faculty of Information Technology of MSUPE.

**Keywords:** higher education, methods of teaching mathematics, analytical geometry, affine geometry, projective geometry, geometric transformations, non-Euclidean geometry, spherical geometry, second-order curves.

### **For citation:**

Kulanin Y.D., Stepanov M.E. A Computational Experiment in Teaching Higher Mathematics. Geometry. *Modelirovanie i analiz dannykh = Modelling and Data Analysis*, 2024. Vol. 14, no. 4, pp. 138–162. DOI: <https://doi.org/10.17759/mda.2024140410> (In Russ., abstr. in Engl.).

### **References**

1. Kulanin E.D., Nurkaeva I.M. On two geometric extremum problems. *Mathematics at school*. 2019. No. 4. pp. 35–40.
2. Kulanin E.D., Nurkaeva I.M. Once again about the Mavlo task. *Mathematics at school*. 2020. No. 2. pp. 76–79.
3. Kulanin E.D., Stepanov M.E., Nurkaeva I.M. Propaedeutics of solving extreme problems in the school mathematics course. *Data modeling and analysis*. 2019. No. 4. pp. 127–144.
4. Kulanin E.D., Nguyen Wu Quang, Stepanov M.E. Tangible objectivity with computer support. *Data modeling and analysis. Scientific journal*. 2019. No. 4. pp. 145–156.

\****Yevgeny D. Kulanin***, Candidate of Physical and Mathematical Sciences, Professor, Moscow State University of Psychology & Education (MSUPE), Moscow, Russia, ORCID: <https://orcid.org/0000-0001-6093-7012>, e-mail: [lucas03@mail.ru](mailto:lucas03@mail.ru)

\*\****Mikhail E. Stepanov***, Candidate of Pedagogical Sciences, Associate Professor, Moscow State University of Psychology & Education (MSUPE), Moscow, Russia, ORCID: <https://orcid.org/0000-0003-4803-8211>, e-mail: [mestepanov@yandex.ru](mailto:mestepanov@yandex.ru)





5. Kulanin E.D., Stepanov M.E., Nurkaeva I.M. The role of imaginative thinking in scientific thinking. *Data modeling and analysis*. 2020. Vol.10. No. 2 pp. 110–128.
6. Kulanin E.D., Stepanov M.E., Nurkaeva I.M. On various approaches to solving extreme problems. *Data modeling and analysis*. 2020. Vol. 11. No.1. pp. 40–60.
7. Lungu K.N., Norin V.P., Pisman D.T., Shevchenko Yu.A., Kulanin E.D. Collection of problems in higher mathematics with control papers. Moscow, 2013. Volume 2 (8th edition).
8. Stepanov M.E. Some questions of the methodology of teaching higher mathematics. *Data modeling and analysis*. 2017. No.1. pp. 54–94.
9. Kulanin E.D., Stepanov M.E. From the experience of working in remote mode Learning Modeling and data analysis. 2022. Vol. 12. No. 3. pp. 58–70.
10. Kulanin E.D., Stepanov M.E. Comprehensive consideration of mathematical concepts as a methodological technique. *Data modeling and analysis*. 2022. Vol. 12. No.4. pp.67–84.
11. Kulanin E.D., Stepanov M.E. On visualization of solutions to some extreme problems. *Data modeling and analysis*. 2022. Vol.12. No.4. pp. 94–104.
12. Kulanin E.D., Stepanov M.E., Panfilov A.D., Potonyshchev I.S. A systematic approach to the methodology of typhlopedagogy on the example of mathematical analysis problems. 2022. Vol. 12. No. 2. pp. 34–82.
13. Kulanin E.D., Stepanov M.E. Computational experiment in teaching higher mathematics by the example of number theory. *Data modeling and analysis*. 2024. vol. 14. No.1. pp. 170–195.
14. *Encyclopedia of Elementary Mathematics. The fourth book. Geometry.* M., State Publishing House of Physics and Mathematics lit., 1963.
15. Lysenko V.I. Nikolay Ivanovich Fuss, M., Nauka, 1975.

Получена 09.10.2024

Принята в печать 06.11.2024

Received 09.10.2024

Accepted 06.11.2024